AD-A119 646 ARMY MISSILE COMMAND REDSTONE ARSENAL AL PROPULSION D--ETC F/G 19/1

DESIGN OF SOLID PROPELLANT GRAINS USING THE HEWLETT-PACKARD 41C--ETC(U)

MAR 82 J C HODGES

UNCLASSIFIED DRSMI/TR-RK-82-3 SBI-AD-E950 284 NL

1 OF 1
ADA
119646

END
DATE
FILMED
10-82
DTIC

AD A119646

ADE 950 204

(12)

TECHNICAL REPORT RK-82-3

DESIGN OF SOLID PROPELLANT GRAINS USING THE HEWLETT-PACKARD 41C CALCULATOR

James C. Hodges, Jr.
Propulsion Directorate
US Army Missile Laboratory

March 1982

U.S. ARMY MISSILE COMMAND
Redstone Arsenal, Alabama 35898

Approved for public release; distribution unlimited

DISPOSITION INSTRUCTIONS

DESTROY THIS REPORT WHEN IT IS NO LONGER NEEDED. DO NOT RETURN IT TO THE ORIGINATOR.

DISCLAIMER

THE FINDINGS IN THIS REPORT ARE NOT TO BE CONSTRUED AS AN OFFICIAL DEPARTMENT OF THE ARMY POSITION UNLESS SO DESIGNATED BY OTHER AUTHORIZED DOCUMENTS.

TRADE NAMES

USE OF TRADE NAMES OR MANUFACTURERS IN THIS REPORT DOES NOT CONSTITUTE AN OFFICIAL INDORSEMENT OR APPROVAL OF THE USE OF SUCH COMMERCIAL HARDWARE OR SOFTWARE.

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER TR-RK-82-3	2. GOVT ACCESSION NO. AD-A119 646	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) Design of Solid Propellant Grains Using the Hewlett-Packard 41C Calculator		5. TYPE OF REPORT & PERIOD COVERED Technical Report
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) James C. Hodges, Jr.		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS Commander, USAMICOM, DRSMI-RK Redstone Arsenal, AL 35898		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS
11. CONTROLLING OFFICE NAME AND ADDRESS Commander, USAMICOM, ATTN: DRSMI-RPT, Redstone Arsenal, AL 35898		12. REPORT DATE March 1982
		13. NUMBER OF PAGES 33
14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office)		15. SECURITY CLASS. (of this report) UNCLASSIFIED
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public ~~distribution~~ RELEASE, distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Solid Propellant Grain Design
Wagon Wheel Grain Design
Modified Wagon Wheel Grain Design
Star Grain Design

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

This report describes the solution of a typical solid propellant grain design requirement using the Hewlett-Packard 41C hand held calculator and a program for the solution of Wagon Wheel, Modified Wagon Wheel and Star Grain designs. A grain design for an experimental rocket, PLUME is given. The HP41C Program is given in Appendix A.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

THE DESIGN OF SOLID PROPELLANT GRAINS USING THE HEWLETT PACKARD 41C CALCULATOR

J. C. HODGES

The design of a solid propellant rocket motor grain is in many ways as much an art as it is a science. Typically, the grain designer has many years of design experience to call upon in designing a grain for a specific application and over the years very sophisticated computer codes have been written to evaluate a given grain design. However, the initial design had to come from the tedious and often time consuming method of graphically laying out a design based on the designer's experience. There have been computer calculated curves published that aid the designer in his choice but for the most part these approaches virtually ignore propellant characteristics and dwell on such parameters as loading fraction. Consequently, it would be advantageous for the designer to have a simple method that would give him a quick solution to his design problem. With this in mind, a program has been written for the H/P 41C Programmable Calculator that will solve for three basic solid propellant grain designs - Standard Wagon Wheel, Modified Wagon Wheel ($W_1=W_2=W_3$), and Star. The program uses ballistic requirements and propellant characteristics selected by the design engineer and calculates a design output that can be used as an excellent first order solution for input to more sophisticated computer codes such as the Solid Propellant Rocket Motor Internal Ballistics Computer Program.[1]

Using the first order solution as a "seed" will greatly simplify the computer code input and reduce computer run time. For simplification, only cylindrical grains without forward or aft end domes are considered, however the engineer does have the option of considering both ends of the grain inhibited, one end of the grain inhibited or botn ends uninhibited. Stable combustion and non-erosive burning are assumed and the number of spokes or star points have been limited to what has been considered practical and useful geometric limits. The equations for the star grain have been established to achieve as close to neutral surface versus time as is feasible within these limits.

User inputs for this program are: grain outside diameter, web burn time, propellant burn rate and K_n ratio at the selected operating pressure, propellant density, propellant expected delivered impulse and the total impulse required by the mission. Program outputs include predicted ballistic parameters, average thrust, etc., and a geometric description of the grain as shown in Figures A, B, and C.

1. " Solid Propellant Rocket Motor Internal Ballistics Computer Program", Report Number RK-TR-67-7, US Army Missile Command, Redstone Arsenal, Alabama.

The equations used in this program are based on a paper by Max W. Stone.[2] The program format has been fashioned after that of the Hewlett/Packard format for the submission of user programs to the H/P Users Library because this format is considered the standard for H/P 41C users. Equipment necessary to use the "SPGD" program are: a H/P 41C with quad memory module or a 41CV, and a H/P 82143A Peripheral Printer. Any suggestions for improvement of the program will be welcomed by the author. One word of caution, negative outputs indicate geometric limits have been exceeded or incorrect data has been entered.

An example of a grain designed using this program is the PLUME grain. PLUME is an experimental rocket for the evaluation of the interaction between the rocket exhaust and the command guidance link. The desired ballistic parameters for PLUME were a velocity of 4000 fps at the end of a 1 second boost, and to maintain 4000 fps for 2 seconds using a sustainer. The motor I.D. was 6.4 inches and minimum smoke propellant had to be used in the boost and sustain phase. The total inert weight of the missile was 57.3 lb.

Preliminary trajectory analysis[3] using the H/P 41C indicated a boost total impulse of 13,500 lb-sec would meet the desired end-of-boost velocity. Drag at the end of boost was 925 lbf, therefore the sustain total impulse would be 1850 lb-sec. A minimum smoke propellant, Propulsion Directorate formulation B-59, of the lacquer type was chosen. Its ballistic characteristics at the selected operating pressure of 2000 psi are: burn rate-0.47 ips, Kn ratio-450 and delivered I_{sp}-245 lb-sec per lb. The density of this formulation is 0.06 lb per cubic inch.

A two part booster grain was selected for the first analysis - a forward boost grain that would burn into a "cylindrical port" sustainer and an aft boost grain that would burn to the case wall during boost burning. The aft section of the boost grain was considered to have one end burning. The forward grain section ends were considered to be inhibited. The "Solid Propellant Grain Design" program was used to determine the geometric and ballistic parameters of the two booster grains. The data inputs and output is shown in Table 1.

The nozzle throat area is sized by the booster requirements and the sustainer ballistic characteristics are "backed into." Propellant experimental data indicated the minimum desirable chamber pressure for stable combustion would probably be 500 psi. The initial burning surface for the sustainer would be the final burning surface for the booster and can be found in the printout (Table 1). This surface will change very quickly as the slivers of the aft booster burn away, however the slivers from the forward grain will remain as cusps on a basically cylindrical port grain.

2. "A Practical Mathematical Approach to Grain Design," Jet Propulsion, Volume 28, Number 1, January 1958.

3. "Multi-Stage Trajectory Analysis" by Karl L. Remmler, Hewlett/Packard Users Library Number 00617C.

Assuming that the sustainer burning surface is 898 square inches subsequent to aft sliver burnout, the motor K_n will be 207. This K_n will cause the chamber pressure to be approximately 500 psi and hence a propellant burn rate of 0.29 ips. The final sustainer burn surface will be approximately 495 square inches giving a chamber pressure of less than 200 psi. This gives an average sustainer burn surface of 696 square inches which calculates to an average thrust of 2755 lbf. It will be noted that this is more thrust than required to overcome drag, but is the minimum condition for sustainer operations using the required minimum smoke propellant. A cross-section sketch of the grain is shown in Figures 1 and 2 and the predicted booster interior ballistics in Table 1. Propellant distribution is shown in Table 2. Table 3 lists the inputs and outputs of a two dimension trajectory analysis of the PLUME missile using the 41C. This design was then used as a "seed" for the Boeing Solid Propellant Rocket Motor Internal Ballistic Computer Program. There was a very good correlation between the ballistics predicted by the H/P 41C and the large-scale computer code.

TABLE 1

PLUME BOOST GRAIN INPUT/OUTPUT

AFT BOOST GRAIN

```
SPGD GRAIN
GRAIN O.D. ?
             6.40    RUN
BURN TIME ?
             1.00    RUN
PROP K ?
           450.00    RUN
PROP DENSITY
              .06    RUN
BURN RATE ?
              .47    RUN
DEL ISP ?
           245.00    RUN
TOT ISP ?
         7,500.00    RUN
LBL A,B, OR C?
                   XEQ B
N=? 4 TO 16
             8.00    RUN
Z=?
              .25    RUN
MOD W.W. GRN.                   ONE END BURNING

AVG F = 7,500.00 LBF            INL Ab= 1,095.23 SQ IN
NET I TOT =7,500.00 SEC         FNL Ab= 1,077.83 SQ IN
NET PROP WT = 30.61 LB          P.R.= 0.98
TOT PROP WT = 33.95 LB          THROAT A= 2.41 SQ IN
BURN TIME = 1.00 SEC            PORT/THROAT= 4.02
INL Ab = 1,072.75 SQ IN
FNL Ab = 1,098.33 SQ IN
P.R. = 1.02
L.F. = 0.70
SLVR FRAC = 0.07
GRAIN L = 25.18 IN
GRAIN O.D. = 6.40 IN
WEB = 0.47 IN
STRESS R = 0.19 IN
SYM NR = 8.00
BETA = 45.00 DEG
ALPHA = 29.88 DEG
THETA = 105.12 DEG
GAMMA = 15.12 DEG
K = 0.96 IN
M = 0.66 IN
L = 1.63 IN
Z = 0.25 IN
PORT A = 9.70 SQ IN
THROAT A = 2.41 SQ IN
PORT/THROAT = 4.02
```

FORWARD BOOST GRAIN

```
GRAIN O.D. ?
             5.24    RUN
BURN TIME ?
             1.00    RUN
PROP K ?
           450.00    RUN
PROP DENSITY
              .06    RUN
BURN RATE ?
              .47    RUN
DEL ISP ?
           245.00    RUN
TOT ISP ?
         6,000.00    RUN
LBL A,B, OR C?
                   XEQ B
N=? 4 TO 16
             8.00    RUN
Z=?
              .25    RUN
MOD W.W. GRN.

AVG F = 6,000.00 LBF
NET I TOT =6,000.00 SEC
NET PROP WT = 24.49 LB
TOT PROP WT = 28.19 LB
BURN TIME = 1.00 SEC
INL Ab = 853.81 SQ IN
FNL Ab = 883.05 SQ IN
P.R. = 1.03
L.F. = 0.76
SLVR FRAC = 0.10
GRAIN L = 28.79 IN
GRAIN O.D. = 5.24 IN
WEB = 0.47 IN
STRESS R = 0.16 IN
SYM NR = 8.00
BETA = 45.00 DEG
ALPHA = 26.66 DEG
THETA = 108.34 DEG
GAMMA = 18.34 DEG
K = 0.40 IN
M = 0.66 IN
L = 1.07 IN
Z = 0.25 IN
PORT A = 5.25 SQ IN
THROAT A = 1.93 SQ IN
PORT/THROAT = 2.72
```

TABLE 2

PLUME PROPELLANT WEIGHT DISTRIBUTION (LBS)

	TOTAL LOADED	NET	SLIVER
Fwd Boost Wt =	28.19	24.49	3.70
Aft Boost Wt =	33.95	30.61	3.34
TOTAL BOOST WT =	62.14	55.10	7.04*
Sustain Cycl Wt =	18.63	16.38	2.25
Boost Slvr Wt =	7.04*	7.04*	-0-
TOTAL SUSTAIN WT =	25.67	23.42	2.25
G. T. Propellant Wt =	87.81	78.52	2.25
Inert Weight =	57.31		
MISSILE WT =	145.12		

*NOTE: Boost sliver consumed during sustain.

Total Grain Length = 25.18 in Aft Boost + 28.79 Fwd Boost = 53.97 in

TABLE 3

PLUME TRAJECTORY ANALYSIS

BOOST INPUT/OUTPUT

A=	.230	RUN
WP=?	55.100	RUN
VAC. ISP=?	270.000	RUN
T0=?	0.000	RUN
TB=?	1.000	RUN
TF=?	1.000	RUN
ALPHA=?	0.000	RUN
AEXIT=?	.230	RUN
N=?	10.000	RUN
M=?	5.000	RUN
W0=?	145.120	RUN
V=?	0.000	RUN
PATH ANGLE=?	3.500	RUN
RANGE=?	0.000	RUN
ALT=?	1.000	RUN
		XEQ A

TIME=0.500 SEC
VEL=1,754.235 FPS
PATH ANGLE=2.281 DEG
RANGE=0.129 KM
ALT=19.658 FT
Q=3,659.912 PSF
T=14,386.160 LBS
DRAG=337.108 LBS
WEIGHT=117.570 LBS

TIME=1.000 SEC
VEL=3,643.860 FPS
PATH ANGLE=1.935 DEG
RANGE=0.554 KM
ALT=69.674 FT
Q=18,127.872 PSF
T=0.000 LBS
DRAG=958.964 LBS
WEIGHT=90.020 LBS

SUSTAIN INPUT/OUTPUT

A=	.230	RUN
WP=?	23.420	RUN
VAC. ISP=?	265.000	RUN
TB=?	3.000	RUN
TF=?	3.000	RUN
ALPHA=?	0.000	RUN
AEXIT=?	.230	RUN
N=?	20.000	RUN
M=?	5.000	RUN
WD=?	0.000	RUN
		XEQ A

TIME=1.500 SEC
VEL=3,957.099 FPS
PATH ANGLE=1.698 DEG
RANGE=1.134 KM
ALT=129.790 FT
Q=18,719.448 PSF
T=2,614.296 LBS
DRAG=990.259 LBS
WEIGHT=84.165 LBS

TIME=2.000 SEC
VEL=4,262.866 FPS
PATH ANGLE=1.479 DEG
RANGE=1.762 KM
ALT=186.652 FT
Q=21,687.893 PSF
T=2,615.313 LBS
DRAG=1,147.290 LBS
WEIGHT=78.310 LBS

TIME=2.500 SEC
VEL=4,558.518 FPS
PATH ANGLE=1.276 DEG
RANGE=2.436 KM
ALT=239.585 FT
Q=24,762.021 PSF
T=2,616.258 LBS
DRAG=1,309.911 LBS
WEIGHT=72.455 LBS

TIME=3.000 SEC
VEL=4,778.121 FPS
PATH ANGLE=1.086 DEG
RANGE=3.153 KM
ALT=287.958 FT
Q=27,890.223 PSF
T=0.000 LBS
DRAG=1,475.393 LBS
WEIGHT=66.600 LBS



Figure 1. PLUME aft boost grain



Figure 2. PLUME forward boost grain and sustain grain

APPENDIX A

SOLID PROPELLANT GRAIN DESIGN

PROGRAM FOR H/P 41C

GLOSSARY OF TERMS

SYMBOL	DESCRIPTION	UNITS
A_b	Area of Burning Surface	sq in
A_c	Cross-section Area of Case	sq in
A_g	Cross-section Area of Grain	sq in
A_{slvr}	Cross-section Area of Grain Sliver	sq in
A_t	Motor Nozzle Throat Area	sq in
Avg	Average	
F	Motor Thrust	Lbf
FNL	Final Conditions at burn-out	
INL	Initial Conditions at ignition	
I_{sp}	Propellant Specific Impulse	Lb_{f-sec}/Lb_m
I_T	Motor Total Impulse	Lb_{f-sec}
K	Length of Short Spoke	in
K_n	Ratio of A_b to A_t	
L	Length of Long Spoke	in
L_g	Length of Grain	in
L'_g	Length of Grain with ends burning.	in
L.F.	Loading Fraction, Ratio of A_g to A_c	
m	Length of Spoke Face	in
N	Symentry Number, Number of Spokes or Star Points	
O.D. or D.	Outside Diameter of Grain	in
PROP	Propellant	
P.R.	Progressivity Ratio, Ratio of FNL A_b to INL A_b	
$\dot{r}$	Burning Rate of Propellant	in/sec
r	Stress Relief Radius of Grain	in

S	Burning Periphery of Grain Cross-section	in
S.F.	Sliver Fraction, Ratio of A_{slvr} to A_c	
t_b	Time required to burn web distance	sec
W	Grain Web	in
Wt	Propellant Weight	Lb_m
Z	Distance between parallel sides of spokes and distance between spoke face and center line (See Figures)	in

SYMBOL	DEFINITION	
α	Alpha, Angle Locating r (see figures)	Deg
β	Beta, Angle Defining Propellant Sector	Deg
γ	gamma, Angle Locating r (see figures)	Deg
ρ	rho, Propellant Density	Lb_m/cu in
θ	Theta, Angle Defining Arc of r	Deg

PROGRAM DESCRIPTION

(EQUATIONS)

STANDARD WAGON WHEEL, FIGURE A

1. N = Number of spokes or symmetry number

2. $\beta = \frac{\pi}{N}$

3. $\gamma = \text{SIN}^{-1} \left[\frac{r+w}{\frac{D}{2} - r - w} \right]$

4. $\alpha = \beta - \gamma$

5. $\theta = \frac{\beta}{2} + \beta - \alpha$

6. Maximum allowable web = (Ø.2)(O.D.)

7. $m = \frac{W}{\text{SIN } \beta}$

8. L minimum = m

9. $\text{L maximum} = \frac{r+w}{\text{TAN } \gamma} - \frac{w}{\text{TAN } \beta}$

10. $L = \frac{(\text{SIN } \beta)(\text{L max}) - Z}{\text{SIN } \beta}$

11. $\text{INL Ab} = \text{Lg2N} \left[L + m + r\theta + \left(\frac{D}{2} - W\right)(\alpha) \right]$

12. $\text{Ag} = N \left[\alpha(Dw - W^2) + 2LW + \frac{w^2}{\text{TAN } \beta} + \frac{D^2 \gamma}{4} - r^2\theta - \frac{(r+W)^2}{\text{TAN } \gamma} \right]$

13. $\text{Aslvr} = N \left[\frac{D^2 \gamma}{4} - (r+W)^2 \theta - \frac{(r+W)^2}{\text{TAN } \gamma} \right]$

14. $\text{L.F.} = \frac{\text{Ag}}{\text{Acase}}$

15. $\text{S.F.} = \frac{\text{Aslvr}}{\text{Acase}}$

16. Weight of slvr = (Aslvr)(ρ)(Lg)

17. $\text{FNL Ab} = (\text{Lg})(2N) \left[L = \left(\frac{D}{2}\right)(\alpha) + (W+r)\theta + \frac{W}{\text{TAN } \beta} - m \right]$

18. $\text{P.R.} = \frac{\text{FNL Ab}}{\text{INL Ab}}$



Figure A. Sector of a standard wagon wheel configuration

MODIFIED WAGON WHEEL, (WEB_1 = WEB_2 = WEB_3), FIGURE B.

1. N = Number of spokes of symmetry number

2. $\beta = \frac{2\pi}{N}$

3. $\gamma = SIN^{-1}\left[\frac{r+W}{\frac{D}{2} - r - W}\right]$

4. $\alpha = \beta - \gamma$

5. $\theta = \frac{\pi}{2} + \gamma$

6. $m = \frac{W}{SIN\ \beta}$

7. $K = L - m$

8. $Lmax = \frac{r+W}{TAN\ \gamma} - \frac{W}{TAN\ \beta}$

9. $L = \frac{(SIN\ \beta)(Lmax) - Z}{SIN\ \beta}$

10. $INL\ Ab = (Lg)(2N)(L) + N\left[m + 2r\theta + \frac{D}{2} - W)(\alpha+\gamma)\right]$

11. $Ag = N\left[\frac{W^2}{TAN\ \beta} + 2LW - mW - r^2\theta + \frac{D^2\beta}{8} - \frac{(r+w)^2}{TAN\ \gamma} - \frac{\left(\frac{D}{2} - W\right)^2\ (\alpha - \gamma)}{2}\right]$

12. $Aslvr = N\left[\frac{D^2\gamma}{4} - (r+W)^2\ \theta - \frac{(r+W)^2}{TAN\ \gamma}\right]$

13. $L.F. = \frac{Ag}{Acase}$

14. $S.F. = \frac{Aslvr}{Acase}$

15. $Wt.\ slvr = (Aslvr)(\rho)(Lg)$

16. $FNL\ Ab = (Lg)(N)\left[2L - 3m + 2(r+W)\theta + \frac{2W}{TAN\beta} + (\frac{D}{2})(\alpha-\gamma)\right]$

17. $P.R. = \frac{FNL\ Ab}{INL\ Ab}$



Figure B. Sector of a modified wagon wheel configuration, $W_1=W_2=W_3$

STAR, FIGURE C.

1. N = Number of Star Points or symmetry number

2. $\beta = \frac{\pi}{N}$

3. $\theta = \beta + \mathrm{SIN}^{-1}\left[\frac{(\frac{D}{2} - r - W)\ (\mathrm{SIN}\ \beta)}{r + w}\right]$

4. $\alpha = \mathrm{TAN}\ (\alpha - \beta)$ for maximum neutrality

5. Maximum Web = (Ø.33)(O.D.)

6. $\mathrm{INL\ Ab} = (Lg)(2N)\left[r(\alpha - \mathrm{TAN}\ \alpha) + \frac{(\frac{D}{2} - r - W + \frac{r}{\mathrm{COS}\ \alpha})\ \mathrm{SIN}\ \beta}{\mathrm{COS}\ (\alpha - \beta)}\right]$

7. $\mathrm{FNL\ Ab} = (Lg)(2N)\left[(r+W)\theta\right]$

8. $Ag = \frac{\pi D^2}{4} - N\left[\frac{(\frac{D}{2} - r - W + \frac{r}{\mathrm{COS}\ \alpha})^2\ \mathrm{SIN}\ \beta\ \mathrm{COS}\ \alpha}{\mathrm{COS}\ (\alpha - \beta)} + r^2(\alpha - \mathrm{TAN}\ \alpha)\right]$

9. $Aslvr = \frac{\pi D^2}{4} - N\left[(r+W)^2\theta + (r+W)(\frac{D}{2} - r - W)\ \mathrm{SIN}\theta\right]$

10. Iteration for $\alpha = \mathrm{TAN}\ (\alpha - \beta)$:

$$\text{Change in } \alpha = \frac{\alpha - \mathrm{TAN}\ (\alpha - \beta)}{\frac{1}{\mathrm{COS}^2\ (\alpha - \beta)} - 1}$$

CORRECTIONS FOR END BURN, ALL CASES

1. New INL Ab = Old INL Ab + (Ag * 2), for two ends

2. New FNL Ab = $(\frac{\mathrm{OLD\ FNL\ Ab}}{Lg})(Lg - 2W)$, for two ends

3. New P.R. = $\frac{\mathrm{New\ FNL\ Ab}}{\mathrm{New\ INL\ Ab}}$

4. New $A_t = \frac{\mathrm{Avg\ Ab}}{Kn}$



Figure C. Sector of Star Design, side burns out at web

GENERAL BALLISTIC EQUATIONS

1. $\text{AVG F} = \dfrac{\text{NET Itotal}}{\text{tb}}$

2. $\text{NET PROP. WT.} = (\text{Lg})(\text{Ag-Aslvr})(\rho)$

3. $\text{TOTAL PROP. WT.} = (\text{Lg})(\text{Ag})(\rho)$

4. $\text{Itotal} = (\text{NET PROP. WT.})(\text{Isp})$

5. $\text{Lg} = \dfrac{\dfrac{\text{Itotal}}{(\rho)(\text{Isp})}}{\text{Ag - Aslvr}}$

6. $W = (\dot{r})(\text{tb})$

7. $K_n = \dfrac{\text{Ab}}{\text{At}}$

8. $A_t = \dfrac{\text{INL Ab + FNL Ab}}{(2)(\text{Kn})}$

PROGRAM DESCRIPTION

SAMPLE PROBLEM:

GIVEN: GRAIN O.D. = 4.25 IN.
BURN TIME = 1.0 SEC
TOTAL IMPULSE = 7ØØØ LB-SEC

REQUIRED: CALCULATE A STD W. W. GRAIN WITH BOTH ENDS BURNING, A MOD W.W. AND A STAR GRAIN WITH NO ENDS BURNING.

SELECTED: CHAMBER PRESSURE = 2ØØØ PSI
PROPELLANT Kn = 46Ø @ 2ØØØ PSI
PROPELLANT BURN RATE = Ø. 5 IPS @ 2ØØØ PSI
PROPELLANT DENSITY = Ø.Ø62 LB/IN^3
PROPELLANT DEL. ISP = 235 LB-SEC/LB

SOLUTION:

INPUT	FUNCTION	DISPLAY	COMMENTS
	PRINTER: ON/NORM		
	(XEQ) SIZE Ø37		
	(EQ) SPGD	READY	
	R/S	GRAIN O.D. ?	ENTER GRAIN O.D.
4.25	R/S	BURN TIME ?	ENTER BURN TIME
1.Ø	R/S	PROP K ?	ENTER PROPELLANT Kn
46Ø	R/S	PROP DENSITY	ENTER PROPELLANT DENSITY
.Ø62	R/S	BURN RATE ?	ENTER PROPELLANT BURN RATE
.5	R/S	DEL ISP ?	ENTER EXPECTED DELIVERED ISP
235	R/S	TOT ISP ?	ENTER TOTAL IMPULSE REQ'D
7ØØØ	R/S	LBL A.B. OR C ?	SELECT LBL A FOR STD W.W.
	(A)	N=? 2 TO 16	SELECT SYMENTRY NR
5.	R/S	Z=?	SELECT VALUE FOR Z
.2	R/S	STD W.W. GRAIN	OUTPUT PRINTED
		NEXT CMD	SELECT LBL D FOR END BRN
	(D)	1 END: SFØ2	SELECT 2 END BRN (NO FLAG)
	R/S		CORRECTION FOR END BRN PRINTED
		NEXT CMD	SELECT LBL B FOR NDD W.W.
		N=? 4 TO 16	SELECT SYMENTRY NR (MULT OF 2)
6.	R/S	L=?	SELECT VALUE FOR Z
.2	R/S	MOD W.W. GRN	OUTPUT PRINTED
		NEXT CMD	SELECT LBL C FOR STAR
	(C)	N=? 4 TO 10	SELECT SYMENTRY NR
1Ø.	R/S	STAR	OUTPUT PRINTED
		NEXT CMD	SELECT LBL E TO EXIT PROGRAM
	(E)	Ø.ØØ	CALCULATOR RTN TO NORMAL

SAMPLE PROBLEM INPUT/OUTPUT

```
          XEQ "SPGD"
SPGD READY
                    RUN
SPGD GRAIN
GRAIN O.D. ?
          4.25      RUN
BURN TIME ?
          1.00      RUN
PROP K ?
        460.00      RUN
PROP DENSITY
          .062      RUN
BURN RATE ?
           .50      RUN
DEL ISP ?
        235.00      RUN
TOT ISP ?
      7,000.00      RUN
LBL A,B, OR C?
                  XEQ A
N=? 2 TO 16
          5.00      RUN
Z=?
           .20      RUN
STD W.W. GRN.

AVG F = 7,000.00 LBF
NET I TOT =7,000.00 SEC
NET PROP WT = 29.79 LB
TOT PROP WT = 34.93 LB
BURN TIME = 1.00 SEC
INL Ab = 937.79 SQ IN
FNL Ab = 983.97 SQ IN
P.R. = 1.05
L.F. = 0.74
SLVR FRAC = 0.11
GRAIN L = 53.41 IN
GRAIN O.D. = 4.25 IN
WEB = 0.50 IN
STRESS R = 0.13 IN
SYM NR = 5.00
BETA = 36.00 DEG
ALPHA = 11.23 DEG
THETA = 114.77 DEG
M = 0.85 IN
L = 0.33 IN
Z = 0.20 IN
PORT A = 3.64 SQ IN
THROAT A = 2.09 SQ IN
PORT/THROAT = 1.74
```

```
                    XEQ D
                       W/
1 END: SF02
                      RUN
TWO END BURNING

INL Ab= 958.89 SQ IN
FNL Ab= 965.54 SQ IN
P.R.= 1.01
THROAT A= 2.09 SQ IN
PORT/THROAT= 1.74

                    XEQ B
N=? 4 TO 16
          6.00      RUN
Z=?
           .20      RUN
MOD W.W. GRN.

AVG F = 7,000.00 LBF
NET I TOT =7,000.00 SEC
NET PROP WT = 29.79 LB
TOT PROP WT = 35.49 LB
BURN TIME = 1.00 SEC
INL Ab = 907.25 SQ IN
FNL Ab = 1,014.51 SQ IN
P.R. = 1.12
L.F. = 0.82
SLVR FRAC = 0.13
GRAIN L = 49.34 IN
GRAIN O.D. = 4.25 IN
WEB = 0.50 IN
STRESS R = 0.13 IN
SYM NR = 6.00
BETA = 60.00 DEG
ALPHA = 35.23 DEG
THETA = 114.77 DEG
GAMMA = 24.77 DEG
K = 0.26 IN
M = 0.58 IN
L = 0.84 IN
Z = 0.20 IN
PORT A = 2.59 SQ IN
THROAT A = 2.09 SQ IN
PORT/THROAT = 1.24
```

```
                    XEQ C
N=?, 4 TO 10
          10.00     RUN
STAR GRN

AVG F = 7,000.00 LBF
NET I TOT =7,000.00 SEC
NET PROP WT = 29.79 LB
TOT PROP WT = 34.49 LB
BURN TIME = 1.00 SEC
INL Ab = 960.87 SQ IN
FNL Ab = 961.65 SQ IN
P.R. = 1.00
L.F. = 0.59
SLVR FRAC = 0.08
GRAIN L = 67.01 IN
GRAIN O.D. = 4.25 IN
WEB = 0.50 IN
STRESS R = 0.13 IN
SYM NR = 10.00
BETA = 18.00 DEG
ALPHA = 67.80 DEG
THETA = 65.52 DEG
PORT A = 5.88 SQ IN
THROAT A = 2.09 SQ IN
PORT/THROAT = 2.82

                    XEQ E
```

USER INSTRUCTIONS

(HP 41C SIZE 037)

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	LOAD PROGRAM			
2	PRINTER ON, MODE: NORM			
3	INITIALIZE PROGRAM		(XEQ) SPGD	SPGD READY
4	PRESS R/S		R/S	GRAIN O.D. ?
	• ENTER GRAIN O.D. (INCHES)	NN	R/S	BURN TIME ?
	• ENTER BRN TIME (SEC)	NN	R/S	PROP K ?
	• ENTER PROP Kn	NN	R/S	PROP DENSITY
	• ENTER PROP DENSITY (LB/IN^3)	NN	R/S	BURN RATE ?
	• ENTER PROP BRN RATE (IPS)	NN	R/S	DEL ISP ?
	• ENTER PROP DEL ISP (LB-SEC/LB)	NN	R/S	TOT ISP ?
	NOTE: PROP Kn, BRN RATE, AND DEL ISP MUST BE AT SAME CHAMBER PRESSURE			
	• ENTER REQ'D TOTAL ISP (LB-SEC)	NN	R/S	LBL A,B OR C ?
5	SELECT TYPE GRAIN:			N = ? N TO N
	• STD. W.W.		(A)	
	• MOD. W.W.		(B)	
	• STAR		(C)	
	(SEE FIGURES A, B, AND C)			
6	ENTER SYMENTRY NUMBER	N	R/S	Ƶ = ?
	• N MUST BE WITHIN LIMITS SHOWN. MOD. W.W. N MUST BE MULTP OF 2.			
7	ENTER SELECTED VALUE FOR Ƶ (IN)	N	R/S	NEXT CMD
	NOTE: NOT APPLICABLE TO STAR			
8	SELECT NEXT CMD			
	• SELECT DIFFERENT GRN TYPE		(A) (B) (C)	(SEE STEP 6)
	• SELECT CORRECTION FOR END BRN		(D)	1 END: SF02
	• FOR ONE END BRN SF02		SF02	
			R/S	NEXT CMD
	• FOR TWO END BRN		R/S	NEXT CMD
	• TO TERMINATE PROGRAM		(E)	0.0 0
	• RETURNS CALCULATOR TO NORMAL MODE			
	• TO CHANGE INPUTS, ENTER NEW VALUE IN PROPER REG		(F)	LBL A.B. OR C

USER INSTRUCTIONS

(HP 41C SIZE 037)

USER MESSAGES

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	• KEY SAME GRN LBL. INPUT		(A) (B) OR (C)	LOWER N OR D
	SMALLER N VALUE OR	N		
	• STORE SMALLER O.D. VALUE IN			
	R00 XEQ (F)	N	(F)	
2	• STORE SMALLER BRN TIME R02	N	(F)	REDUCE WEB
	XEQ (F) OR			
	• STORE SMALLER BRN RATE R05			
	CHANGE Kn AND DEL ISP AS REQ'D			
	Kn IN R03, DEL ISP IN R06	NN	(F)	
3				REDUCE Ƶ
	• KEY IN LOWER VALUE FOR Ƶ	N	R/S	Ƶ = ?
4				REDUCE N
	• KEY IN LOWER VALUE FOR N	N	R/S	N = ?
5	• KEY IN SAME TYPE GRN		(A) (B) OR (C)	INC: N OR
	REDUCE VALUE FOR N, OR	N	R/S	DEC: D
	• STORE REDUCED O.D. IN R00	NN	(F)	LBL A,B, OR
	AND SELECT SAME TYPE GRN LBL		(A) (B) OR (C)	C

PROGRAM LISTING

LINE	KEY ENTRY	COMMENTS
01	◆LBL "SPGD"	
02	FIX 2	
03	SF 27	
04	RAD	
05	"SPGD GRAIN"	
06	AVIEW	
07	◆LBL 00	DATA INPUT
08	"GRAIN O.D. ?"	
09	PROMPT	
10	STO 00	
11	"BURN TIME ?"	
12	PROMPT	
13	STO 02	
14	"PROP K ?"	
15	PROMPT	
16	STO 03	
17	"PROP DENSITY"	
18	PROMPT	
19	STO 04	
20	"BURN RATE ?"	
21	PROMPT	
22	STO 05	
23	"DEL ISP ?"	
24	PROMPT	
25	STO 06	
26	"TOT ISP ?"	
27	PROMPT	
28	STO 07	
29	◆LBL F	CAL WEB
30	RCL 00	r, A_C
31	X↑2	
32	PI	
33	*	
34	4	
35	/	
36	STO 18	A_C
37	RCL 00	
38	.03	
39	*	
40	STO 12	r
41	RCL 00	
42	.2	
43	*	
44	STO 15	MAX W FOR WW GRN
45	RCL 02	
46	RCL 05	
47	*	
48	STO 10	W
49	RCL 15	
50	X<>Y	
51	X>Y?	
52	GTO 17	
53	"LBL A,B, OR C?"	
54	PROMPT	
55	RTN	
56	◆LBL A	CAL STD WW GRN
57	SF 05	
58	CF 06	
59	CF 07	
60	"N=? 2 TO 16"	
61	PROMPT	
62	STO 13	N
63	PI	
64	RCL 13	
65	/	
66	STO 14	BETA
67	XEQ 01	
68	XEQ 02	
69	RCL 21	
70	RCL 19	
71	+	
72	RCL 12	
73	RCL 23	
74	*	
75	+	
76	RCL 00	
77	2	
78	/	
79	RCL 10	
80	-	
81	RCL 17	
82	*	
83	+	
84	RCL 13	
85	2	
86	*	
87	*	
88	STO 11	INL S
89	RCL 00	
90	RCL 10	
91	*	
92	RCL 10	
93	X↑2	

PROGRAM LISTING

LINE	KEY ENTRY	COMMENTS
94	-	
95	RCL 17	
96	*	
97	RCL 21	
98	RCL 10	
99	*	
100	2	
101	*	
102	+	
103	RCL 10	
104	X↑2	
105	RCL 14	
106	TAN	
107	/	
108	+	
109	RCL 00	
110	X↑2	
111	RCL 16	
112	*	
113	4	
114	/	
115	+	
116	RCL 12	
117	X↑2	
118	RCL 23	
119	*	
120	-	
121	RCL 12	
122	RCL 10	
123	+	
124	X↑2	
125	RCL 16	
126	TAN	
127	/	
128	-	
129	RCL 13	
130	*	
131	STO 24	A_g
132	XEQ 04	
133	XEQ 03	
134	RCL 21	
135	RCL 00	
136	2	
137	/	
138	RCL 17	
139	*	
140	+	
141	RCL 10	
142	RCL 12	
143	+	
144	RCL 23	
145	*	
146	+	
147	RCL 10	
148	RCL 14	
149	TAN	
150	/	
151	+	
152	RCL 19	
153	-	
154	RCL 13	
155	*	
156	2	
157	*	
158	RCL 01	
159	*	
160	STO 32	FNL A_b
161	SF 00	
162	"STD W.W, GRN."	
163	AVIEW	
164	ADV	
165	GTO 15	
166	RTN	
167	♦LBL B	CAL MOD
168	SF 06	WW GRN
169	CF 05	
170	CF 07	
171	"N=? 4 TO 16"	
172	PROMPT	
173	STO 13	N
174	PI	
175	2	
176	*	
177	RCL 13	
178	/	
179	STO 14	BETA
180	XEQ 01	
181	RCL 17	
182	RCL 16	
183	X>Y?	
184	GTO 16	
185	XEQ 02	
186	RCL 19	
187	RCL 21	
188	X<Y?	
189	GTO 16	
190	RCL 19	
191	RCL 12	
192	2	
193	*	
194	RCL 23	
195	*	

PROGRAM LISTING

LINE	KEY ENTRY	COMMENTS
196	+	
197	RCL 00	
198	2	
199	/	
200	RCL 10	
201	-	
202	RCL 17	
203	RCL 16	
204	-	
205	*	
206	+	
207	RCL 13	
208	*	
209	RCL 21	
210	RCL 13	
211	*	
212	2	
213	*	
214	+	
215	STO 11	$\bar{I}$NL S
216	RCL 10	
217	X↑2	
218	RCL 14	
219	TAN	
220	/	
221	RCL 10	
222	RCL 21	
223	*	
224	2	
225	*	
226	+	
227	RCL 10	
228	RCL 19	
229	*	
230	-	
231	RCL 12	
232	X↑2	
233	RCL 23	
234	*	
235	-	
236	RCL 00	
237	X↑2	
238	RCL 14	
239	*	
240	8	
241	/	
242	+	
243	RCL 12	
244	RCL 10	
245	+	
246	X↑2	
247	RCL 16	
248	TAN	
249	/	
250	-	
251	RCL 00	
252	2	
253	/	
254	RCL 10	
255	-	
256	X↑2	
257	RCL 17	
258	RCL 16	
259	-	
260	*	
261	2	
262	/	
263	-	
264	RCL 13	
265	*	
266	STO 24	A_g
267	XEQ 04	
268	XEQ 03	
269	RCL 21	
270	2	
271	*	
272	RCL 19	
273	3	
274	*	
275	-	
276	RCL 12	
277	RCL 10	
278	+	
279	2	
280	*	
281	RCL 23	
282	*	
283	+	
284	RCL 10	
285	2	
286	*	
287	RCL 14	
288	TAN	
289	/	
290	+	
291	RCL 00	
292	2	
293	/	
294	RCL 17	
295	RCL 16	
296	-	
297	*	
298	+	
299	RCL 13	

PROGRAM LISTING

LINE	KEY ENTRY	COMMENTS
300	*	
301	RCL 01	
302	*	
303	STO 32	FNL A_b
304	"MOD W.W . GRN."	
305	AVIEW	
306	ADV	
307	GTO 15	
308	RTN	
309	◆LBL C	CAL STAR
310	SF 07	
311	CF 05	
312	CF 06	
313	10	
314	"N=?, 4 TO 10"	
315	PROMPT	
316	X>Y?	
317	GTO 20	
318	STO 13	N
319	PI	
320	RCL 13	
321	/	
322	STO 14	BETA
323	1.87	
324	STO 17	
325	◆LBL 10	ALPHA
326	RCL 17	ITERATION
327	RCL 14	
328	-	
329	TAN	
330	CHS	
331	RCL 17	
332	+	
333	RCL 17	
334	RCL 14	
335	-	
336	COS	
337	X↑2	
338	1/X	
339	1	
340	-	
341	/	
342	X=0?	
343	GTO 09	
344	X>0?	
345	GTO 09	
346	ST+ 17	
347	GTO 10	
348	◆LBL 09	CONT STAR
349	RCL 00	CAL
350	2	
351	/	
352	RCL 12	
353	-	
354	RCL 10	
355	-	
356	RCL 14	
357	SIN	
358	*	
359	RCL 12	
360	RCL 10	
361	+	
362	/	
363	1	
364	X<Y?	
365	GTO 19	
366	RCL Y	
367	ASIN	
368	RCL 14	
369	+	
370	STO 23	THETA
371	RCL 17	
372	X<Y?	
373	GTO 19	
374	RCL 17	
375	TAN	
376	-	
377	RCL 12	
378	*	
379	RCL 00	
380	2	
381	/	
382	RCL 12	
383	-	
384	RCL 10	
385	-	
386	RCL 12	
387	RCL 17	
388	COS	
389	/	
390	+	
391	RCL 14	
392	SIN	
393	*	
394	RCL 17	
395	RCL 14	
396	-	
397	COS	
398	/	
399	+	
400	RCL 13	
401	*	

PROGRAM LISTING

LINE	KEY ENTRY	COMMENTS
402	2	
403	*	
404	STO 11	INL S
405	RCL 00	
406	2	
407	/	
408	RCL 12	
409	-	
410	RCL 10	
411	-	
412	RCL 12	
413	RCL 17	
414	COS	
415	/	
416	+	
417	X↑2	
418	RCL 14	
419	SIN	
420	*	
421	RCL 17	
422	COS	
423	*	
424	RCL 17	
425	RCL 14	
426	-	
427	COS	
428	/	
429	RCL 12	
430	X↑2	
431	RCL 17	
432	RCL 17	
433	TAN	
434	-	
435	*	
436	+	
437	RCL 13	
438	*	
439	CHS	
440	RCL 18	
441	+	
442	STO 24	A_g
443	RCL 12	
444	RCL 10	
445	+	
446	X↑2	
447	RCL 23	
448	*	
449	RCL 12	
450	RCL 10	
451	+	
452	RCL 00	
453	2	
454	/	
455	RCL 12	
456	-	
457	RCL 10	
458	-	
459	*	
460	RCL 23	
461	SIN	
462	*	
463	+	
464	RCL 13	
465	*	
466	CHS	
467	RCL 18	
468	+	
469	STO 25	A_{slvr}
470	XEQ 03	
471	RCL 12	
472	RCL 10	
473	+	
474	RCL 23	
475	*	
476	RCL 13	
477	*	
478	RCL 01	
479	*	
480	2	
481	*	
482	STO 32	
483	SF 01	
484	"STAR GR N"	
485	AVIEW	
486	ADV	
487	GTO 15	
488	RTN	
489	◆LBL 01	CAL ANGLES M. Lmax
490	RCL 12	
491	RCL 10	
492	+	
493	RCL 00	
494	2	
495	/	
496	RCL 12	
497	-	
498	RCL 10	
499	-	
500	/	
501	ASIN	
502	STO 16	GAMMA
503	RCL 14	
504	X<>Y	

PROGRAM LISTING

LINE	KEY ENTRY	COMMENTS
505	X>Y?	
506	GTO 16	
507	CHS	
508	+	
509	STO 17	ALPHA
510	CHS	
511	RCL 14	
512	+	
513	PI	
514	2	
515	/	
516	+	
517	STO 23	THETA
518	RCL 10	
519	RCL 14	
520	SIN	
521	/	
522	STO 19	M
523	RCL 12	
524	RCL 10	
525	+	
526	RCL 16	
527	TAN	
528	/	
529	RCL 10	
530	RCL 14	
531	TAN	
532	/	
533	-	
534	STO 20	L max
535	RTN	
536	♦LBL 02	DATA INPUT
537	"Z=?"	
538	PROMPT	
539	STO 22	Z
540	RCL 20	
541	RCL 14	
542	SIN	
543	*	
544	RCL 22	
545	-	
546	RCL 14	
547	SIN	
548	/	
549	STO 21	CALCULATED
550	0	L
551	RCL 21	
552	X<=Y?	
553	GTO 18	
554	RTN	
555	♦LBL 03	
556	RCL 07	
557	RCL 04	
558	/	
559	RCL 06	
560	/	
561	RCL 24	
562	RCL 25	
563	-	
564	/	
565	STO 01	LENGTH
566	RCL 11	GRN
567	*	
568	STO 09	
569	RTN	
570	♦LBL 04	CAL Aslvr
571	RCL 00	
572	X↑2	
573	RCL 16	
574	*	
575	4	
576	/	
577	RCL 12	
578	RCL 10	
579	+	
580	X↑2	
581	RCL 23	
582	*	
583	-	
584	RCL 12	
585	RCL 10	
586	+	
587	X↑2	
588	RCL 16	
589	TAN	
590	/	
591	-	
592	RCL 13	
593	*	
594	STO 25	A slvr
595	RTN	
596	♦LBL 15	CAL OUTPUT
597	RCL 32	
598	RCL 09	
599	/	
600	STO 33	P.R.
601	RCL 24	
602	RCL 18	
603	/	
604	STO 27	L.F.
605	RCL 25	
606	RCL 18	
607	/	
608	STO 28	S.F.

PROGRAM LISTING

LINE	KEY ENTRY	COMMENTS
609	RCL 18	
610	RCL 24	
611	-	
612	STO 26	A_p
613	RCL 01	
614	RCL 24	
615	*	
616	RCL 04	
617	*	
618	STO 35	TOT. PROP.
619	RCL 01	WT.
620	RCL 24	
621	RCL 25	
622	-	
623	*	
624	RCL 04	
625	*	
626	STO 29	NET PROP.
627	RCL 06	WT.
628	*	
629	STO 30	NET TOTAL
630	RCL 02	IMPULSE
631	/	
632	STO 31	AVG F
633	RCL 09	
634	RCL 32	
635	+	
636	2	
637	/	
638	RCL 03	
639	/	
640	STO 34	A_t
641	GTO 05	
642	RTN	
643	◆LBL 16	USER
644	TONE 8	MESSAGES
645	"LOWER N	
	OR D"	
646	PROMPT	
647	RTN	
648	◆LBL 17	
649	RCL 00	
650	.33	
651	*	
652	RCL 10	
653	X<Y?	
654	GTO 06	
655	TONE 8	
656	"REDUCE	
	WEB"	
657	PROMPT	
658	RTN	
659	◆LBL 06	
660	"USE LBL	
	C"	
661	PROMPT	
662	RTN	
663	◆LBL 18	
664	TONE 7	
665	"REDUCE	
	Z"	
666	AVIEW	
667	GTO 02	
668	RTN	
669	◆LBL 19	
670	TONE 7	
671	"INC:N O	
	R DEC:D"	
672	PROMPT	
673	RTN	
674	◆LBL 20	
675	TONE 7	
676	"REDUCE	
	N"	
677	AVIEW	
678	GTO C	
679	RTN	
680	◆LBL 05	PRINT
681	"AVG F =	OUTPUT
	"	
682	ARCL 31	
683	"⊢ LBF"	
684	PRA	
685	"NET I T	
	OT ="	
686	ARCL 30	
687	"⊢ SEC"	
688	PRA	
689	"NET PRO	
	P WT = "	
690	ARCL 29	
691	"⊢ LB"	
692	PRA	
693	"TOT PRO	
	P WT = "	
694	ARCL 35	
695	"⊢ LB"	
696	PRA	
697	"BURN TI	
	ME = "	
698	ARCL 02	
699	"⊢ SEC"	
700	PRA	
701	"INL Ab	
	= "	

PROGRAM LISTING

LINE KEY ENTRY COMMENTS LINE KEY ENTRY COMMENTS

```
702 ARCL 09
703 "⊢ SQ IN
"
704 PRA
705 "FNL Ab
= "
706 ARCL 32
707 "⊢ SQ IN
"
708 PRA
709 "P.R. =
"
710 ARCL 33
711 PRA
712 "L.F. =
"
713 ARCL 27
714 PRA
715 "SLVR FR
AC = "
716 ARCL 28
717 PRA
718 "GRAIN L
 = "
719 ARCL 01
720 "⊢ IN"
721 PRA
722 "GRAIN O
.D. = "
723 ARCL 00
724 "⊢ IN"
725 PRA
726 "WEB = "
727 ARCL 10
728 "⊢ IN"
729 PRA
730 "STRESS
R = "
731 ARCL 12
732 "⊢ IN"
733 PRA
734 "SYM NR
= "
735 ARCL 13
736 PRA
737 RCL 14
738 R-D
739 "BETA =
"
740 ARCL X
741 "⊢ DEG"
742 PRA
```

```
743 RCL 17
744 R-D
745 "ALPHA =
"
746 ARCL X
747 "⊢ DEG"
748 PRA
749 RCL 23
750 R-D
751 "THETA =
"
752 ARCL X
753 "⊢ DEG"
754 PRA
755 FS?C 00
756 GTO 07
757 FS?C 01
758 GTO 08
759 RCL 16
760 R-D
761 "GAMMA =
"
762 ARCL X
763 "⊢ DEG"
764 PRA
765 RCL 21
766 RCL 19
767 -
768 "K = "
769 ARCL X
770 "⊢ IN"
771 PRA
772♦LBL 07          CONT OUTPUT
773 "M = "          FOR W.W.
774 ARCL 19
775 "⊢ IN"
776 PRA
777 "L = "
778 ARCL 21
779 "⊢ IN"
780 PRA
781 "Z = "
782 ARCL 22
783 "⊢ IN"
784 PRA
785♦LBL 08          CONT OUTPUT
786 RCL 26          FOR MOD
787 "PORT A          WIN
= "
788 ARCL X
789 "⊢ SQ IN
"
```

PROGRAM LISTING

LINE	KEY ENTRY	COMMENTS
790	PRA	
791	RCL 34	
792	"THROAT A = "	
793	ARCL X	
794	"⊢ SQ IN "	
795	PRA	
796	/	
797	"PORT/THROAT = "	
798	ARCL X	
799	PRA	
800	TONE 6	
801	ADV	
802	CF 21	
803	"NEXT CMD"	
804	AVIEW	
805	SF 21	
806	RTN	
807	♦LBL D	CAL END BRN
808	SF 08	
809	"W/"	
810	ACA	
811	"1 END: SF02"	
812	PROMPT	
813	FS? 02	
814	GTO 25	
815	GTO 26	
816	♦LBL 25	
817	"ONE END BURNING"	
818	ACA	
819	PRBUF	
820	GTO 27	
821	♦LBL 26	
822	"TWO END BURNING"	
823	ACA	
824	PRBUF	
825	♦LBL 27	
826	ADV	
827	RCL 09	
828	RCL 24	
829	FS? 02	
830	XEQ 11	
831	2	
832	*	
833	+	
834	STO 09	INL Ab
835	RCL 32	
836	RCL 01	
837	/	
838	RCL 01	
839	RCL 10	
840	2	
841	*	
842	FS?C 02	
843	XEQ 11	
844	-	
845	*	
846	STO 32	FNL Ab
847	RCL 09	
848	+	
849	2	
850	/	
851	RCL 03	
852	/	
853	STO 34	A_t
854	RCL 32	
855	RCL 09	
856	/	
857	STO 33	P.R.
858	"INL Ab= "	
859	ARCL 09	
860	"⊢ SQ IN "	
861	PRA	
862	"FNL Ab= "	
863	ARCL 32	
864	"⊢ SQ IN "	
865	PRA	
866	"P.R.= "	
867	ARCL 33	
868	PRA	
869	"THROAT A= "	
870	ARCL 34	
871	"⊢ SQ IN "	
872	PRA	
873	RCL 26	
874	RCL 34	
875	/	
876	"PORT/THROAT= "	
877	ARCL X	
878	PRA	
879	TONE 8	

PROGRAM LISTING

LINE	KEY ENTRY	COMMENTS	LINE	KEY ENTRY	COMMENTS
880	CF 21				
881	"NEXT CMD"				
882	AVIEW				
883	SF 21				
884	ADV				
885	RTN				
886	♦LBL 11				
887	2				
888	/				
889	RTN				
890	♦LBL E	EXIT PROGRAM			
891	CF 27				
892	DEG				
893	CLX				
894	END				

REGISTER	DATA
ØØ	GRN O.D.
Ø1	GRN LNGTH, Lg
Ø2	BRN TIME, tb
Ø3	Kn
Ø4	DENSITY ρ
Ø5	BRN RATE, $\dot{r}$
Ø6	DEL ISP
Ø7	TOTAL ISP
Ø8	(NOT USED)
Ø9	INL, Ab
1Ø	WEB, w
11	INL Si
12	STRESS RAD, r
13	SYM. NR., N
14	BETA, β
15	MAX WEB
16	GAMMA, γ
17	ALPHA,
18	Acase
19	M
2Ø	MAX L
21	CALCULATED L
22	Z
23	THETA, θ
24	Agrain
25	Aslvr
26	Aport
27	L.F.
28	S.F.
29	NET PROP. WT.
3Ø	NET I TOTAL
31	AVG F
32	FNL Ab
33	P.R.
34	At
35	TOTAL PROP. WT.

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
ØØ	C	STD. W.W. CALCULAT	NOT STD. W.W. CAL
Ø1	C	STAR CALCULATION	NOT STAR CAL
Ø2	C	ONE END BRN	TWO END BRN
27	S	USER MODE	PROGRAM TERMINATED

STATUS

SIZE	037	TOTAL REG.	315	USER MODE ON X OFF ___	
ENG	___	FIX	Ø2		
SCI	___	DEG	___	RAD X	GRAD ___

ASSIGNMENTS

NONE

DISTRIBUTION

ADDRESSEE	COPIES
DRSMI-RKX	1
DRSMI-RKA	1
DRSMI-RKC	1
DRSMI-RKF	1
DRSMK-RKK	5
DRSMI-RKL	1
DRSMI-RKP	1
DRSMI-LP, Mr. Voigt	1
DRSMI-RPR	15
DRSMI-RPT	1

LMED

-8